
Поиск компонент связности графа
Граф задан его матрицей смежности. Требуется определить количество компонент связности этого графа (по материалам главы 3, п. 3.2.3 и 3.4). При этом должны быть конкретно перечислены вершины, входящие в каждую компоненту связности.

Выбор алгоритма поиска компонент связности – произвольный. Например, приветствуется использование одного из видов обхода (поиск в глубину или поиск в ширину по материалам п. 3.4.3).

Пользователю должна быть предоставлена возможность редактировать исходную матрицу, т.е. изменять исходный граф без выхода из программы. Предусмотреть также возможность изменения количества вершин.

При выполнении работы разрешается (даже рекомендуется!) использовать матрицу бинарных отношений из лабораторной работы №1.

Вход  программы: число вершин графа и матрица смежности.

Выход: разбиение множества вершин на подмножества, соответствующие компонентам связности.

Дополнительно: 
Заданный граф рассматривать как ориентированный. Выполнять поиск компонент сильной связности.
Описание работы
Программа работает с неориентированным графом. Для поиска компонент связности введем массив состояний вершин графа, в котором будем отмечать уже просмотренные вершины. Если вершина просмотрена, то в соответствующий элемент массива записываем значение True. В основном цикле поиска компонент связности перебираем все вершины по порядку.

Если вершина еще не просмотрена, т.е. пока не вошла ни в какую компоненту связности, то запускаем алгоритм поиска всей компоненты связности, начиная с этой вершины, т.е. поиска компоненты связности, в которую входит эта вершина. Все вершины компоненты связности  определяем путем поиска в глубину, используя для этого рекурсию. При этом все вершины, принадлежащие этой компоненте связности, помечаем как просмотренные. После нахождения очередной вершины текущей компоненты связности выводим номер вершины на экран.

Описание основных блоков
Процедура SearchForOne - определяет одну компоненту связности посредством поиска в глубину  и выводит ее на экран. Принимает в качестве параметра номер вершины, входящей в эту компоненту связности.

Процедура SearchForAll - ищет все компоненты связности.

Код программы

uses Crt;

const
Nmax = 100; { Максимальное количество вершин графа }

var

A: Array[1..Nmax, 1..Nmax] of Shortint; { Матрица смежности }

N: Integer; { Количество вершин графа }

S: Array[1..Nmax] of Boolean; { Массив состояний для поиска в глубину }

{ определяет одну компоненту связности посредством поиска в глубину

и выводит ее на экран }

procedure SearchForOne(u: Integer);

var

v: Integer;

begin
Write(u, ' '); { Выводим вершину }

S[u] := True; { Помечаем вершину как "просмотренную" }

{ Посещаем смежные с u вершины, которые ещё не были просмотрены }

for v := 1 to N do

if (A[u, v] > 0) and (not S[v]) then SearchForOne(v);

end;

{ Ищет все компоненты связности }

procedure SearchForAll;

var

v, k: Integer;

begin
{ S[v] = True, если вершина уже была просмотрена, и False иначе }

{ Изначально ни одна вершина ещё не просмотрена }

for v := 1 to N do

S[v] :=  False;

k := 0; { Номер компоненты связности }

{ Проходим по всем вершинам, которые ещё не просмотрены }

for v := 1 to N do

if not S[v] then begin

Inc(k);

Write(k, ') ');  { Выводим номер компоненты связности }

SearchForOne(v); { Запускаем поиск всей компоненты связности с вершиной v }

WriteLn;

end;

end;

var

u, v, w: Integer;

mode: String;

c: Char;

k: Integer;

begin

for u := 1 to Nmax do

for v := 1 to Nmax do

A[u, v] := 0;

N := 0;

{ Выводим на экран меню }

ClrScr;

WriteLn('Выполнил);

WriteLn('1 - очистка экрана');

WriteLn('2 - выход');

WriteLn('3 - ввод графа');

WriteLn('4 - вывод компонент связности');

WriteLn('5 - изменение количества вершин');

WriteLn('6 - изменение матрицы смежности');

WriteLn('7 - вывод матрицы смежности');

Window(1, 9, 80, 25);

WriteLn('Выберите режим:');

{ Считываем и выполняем с клавиатуры команды пользователя }

mode := '';

repeat

if mode = '1' then begin

ClrScr;

Write('Режим: ');

end

else if mode = '3' then begin

WriteLn('Введите число вершин:');

{$I-}

ReadLn(N);

{$I+}

while (IOResult <> 0) or (N < 1) or (N > Nmax) do begin

WriteLn('Неправильный ввод. Введите число вершин:');

{$I-}

ReadLn(N);

{$I+}

end;

WriteLn('Введите матрицу смежности построчно (без пробелов нули и единицы)');

for u := 1 to N do begin

for v := u to N do begin

repeat

c := ReadKey;

until c in ['0', '1'];

Write(c);

k := Ord(c) - Ord('0');

A[u, v] := k;

A[v, u] := k;

end;

Writeln;

{ Выводим на экран уже введенные элементы матрицы, т.к. граф

неориентированный, и матрица смежности симметричная }

if u < N then

for v := 1 to u do

Write(A[v, u + 1]);

end;

WriteLn('Граф введен');

WriteLn;

Write('Режим: ');

end

else if mode = '4' then begin

WriteLn('Компоненты связности:');

SearchForAll;

WriteLn;

Write('Режим: ');

end

else if mode = '5' then begin

WriteLn('Введите новое число вершин:');

{$I-}

ReadLn(N);

{$I+}

while (IOResult <> 0) or (N < 1) or (N > Nmax) do begin

WriteLn('Неправильный ввод. Введите число вершин:');

{$I-}

ReadLn(N);

{$I+}

end;

WriteLn('Изменение сделано');

WriteLn;

Write('Режим: ');

end

else if mode = '6' then begin

WriteLn('Введите через пробел номер строки, номер столбца, новое значение элемента');

WriteLn('При этом изменится и второй соответствующий элемент');

Reset(Input);

while not SeekEOLn do begin

{$I-}

ReadLn(u, v, w);

{$I+}

while (IOResult <> 0) or not (w in [0, 1]) or (u < 1) or (v < 1) or

(u > N) or (v > N) do begin

WriteLn('Неправильный ввод. Введите числа снова:');

{$I-}

ReadLn(u, v, w);

{$I+}

end;

A[u, v] := w;

A[v, u] := w;

WriteLn('Введите ещё набор из трех чисел или Enter для окончания ввода');

end;

WriteLn('Изменение сделано');

WriteLn;

Write('Режим: ');

end

else if mode = '7' then begin

WriteLn('Матрица смежности графа:');

for u := 1 to N do begin

for v := 1 to N do

Write(A[u, v]);

WriteLn;

end;

WriteLn;

Write('Режим: ');

end

else if mode = '' then

Write('Режим: ')

else begin

WriteLn('Неправильный режим');

Write('Режим: ');

end;

Reset(Input);

ReadLn(mode);  { Новый режим }

until mode = '2';

end.

Результаты работы программы 

Матрица смежности должна быть квадратной. Каким образом при 4 вершинах у Вас получается 5 элементов в строке?
Далее. При запуске программы должна выводиться информация о типе графа – орграф или неорграф.

